```
http://www.mapleprimes.com/questions/137952-How-To-Find-The-Limit

Reference for asymptotic integrals:
- Malham, An introduction to asymptotic analysis, Chap. 4.1, http://www.ma.hw.ac.uk/~simonm/ae.pdf (used here)
- Avramidi, Notes on Asymptotic Expansions (2001), http://www.nmt.edu/%7Eiavramid/notes/asexp.pdf
- Erdelyi, Asymptotic expansions (1956), Chap. 2.4, Dover Publications
- Wong, Asymptotic approximations of integrals (2001), p.55 ff
- Bender, Orszag, Advanced mathematical methods for scientists and engineers (1978), Chap. 6.4
  with many examples and sketching higher order terms (p. 272 ff)

Reference for asymptotic 2F1:
- Temme, Large Parameter Cases of the Gauss Hypergeometric Function (2002),
http://arxiv.org/abs/math/0205065v1
```

```
> restart; interface(version);
  Digits:=15;
  with(IntegrationTools):
```

Classic Worksheet Interface, Maple 16.00, Windows, Mar 3 2012, Build ID 732982

$$Digits := 15$$

```
> `Assertion:`;
  n*Int((1-(3/26)*x-(37/49)*x^2)^n, x = 0 .. 1):
  Limit(%, n=infinity) = `8` + 2/3;
```

Assertion:

$$\lim_{n \to \infty} n \int_0^1 \left(1 - \frac{3}{26}x - \frac{37}{49}x^2\right)^n dx = 8 + \frac{2}{3}$$

```
>
```

Computing through hypergeometric 2F1

```
> # integrand
  (1-(3/26)*x-(37/49)*x^2); factor(%); #%^1;
  plot(%, x=0..1);
```

$$1 - \frac{3}{26}x - \frac{37}{49}x^2$$

$$-\frac{(74x+91)(13x-14)}{1274}$$




```
> # so do a change of variables:
  t=(1-(3/26)*x-(37/49)*x^2); [solve(%, x)];
  simplify(subs(t=1, %)); # for choosing the correct one
```

$$t = 1 - \frac{3}{26}x - \frac{37}{49}x^2$$

$$\left[-\frac{147}{1924} + \frac{7\sqrt{100489 - 100048\,t}}{1924}, -\frac{147}{1924} - \frac{7\sqrt{100489 - 100048\,t}}{1924}\right]$$

$$\left[0, \frac{-147}{962}\right]$$

```
> n*Int((1-(3/26)*x-(37/49)*x^2)^n, x = 0 .. 1);
  ``=Change(%, x= -147/1924+7/1924*(100489-100048*t)^(1/2), t);
  value(%): expand(%):
  collect(%, hypergeom);
  #A:=op(1,Sol);
  #B:=op(2,Sol);
  #'tmp=A+B'; is(%);
  Sol:=rhs(%):
```

$$n \int_0^1 \left(1 - \frac{3}{26}x - \frac{37}{49}x^2\right)^n dx$$

$$= 182\,n \int_{\frac{165}{1274}}^1 \frac{t^n}{\sqrt{100489 - 100048\,t}}\,dt$$

$$= -\frac{165}{2219}\,\frac{n\,\mathrm{hypergeom}\left(\left[\frac{1}{2}, n+1\right], [2+n], \frac{634920}{4923961}\right) 165^n}{(n+1)\,1274^n} + \frac{182}{317}\,\frac{n\,\mathrm{hypergeom}\left(\left[\frac{1}{2}, n+1\right], [2+n], \frac{100048}{100489}\right)}{n+1}$$

Temme, p. 3/4:

$$\begin{aligned}
{}_2F_1\left(\begin{matrix} a, \beta+\lambda \\ \gamma+\lambda \end{matrix}; z\right) &= (1-z)^{-a}{}_2F_1\left(\begin{matrix} a, \gamma-\beta \\ \gamma+\lambda \end{matrix}; \frac{z}{z-1}\right) \\
&= (1-z)^{-a}\left[1 + \frac{a(\gamma-\beta)}{\gamma+\lambda}\,\frac{z}{z-1} + \frac{(a)(a+1)(\gamma-\beta)(\gamma-\beta+1)}{(\gamma+\lambda)(\gamma+\lambda+1)\,2!}\left(\frac{z}{z-1}\right)^2 \cdots\right] \qquad (2.5) \\
&= (1-z)^{-a}\left[1 + \frac{a(\gamma-\beta)}{\gamma+\lambda}\,\frac{z}{z-1} + \mathcal{O}\left(\lambda^{-2}\right)\right],
\end{aligned}$$

"It is an asymptotic expansion for lambda large, and all fixed z, z <> 1."

```
> (1-z)^(-a)*(1 + a*(gamma - beta)/(gamma+lambda)*z/(z-1) );
  ``=subs(lambda=n, gamma=2, beta=1, %);
  #subs(a=1/2, %);
  rhs(%):
  Limit(%, n=infinity): '%'=value(%);
  ``=subs(a=1/2, rhs(%));
```

$$(1-z)^{(-a)}\left(1 + \frac{a\,(\gamma-\beta)\,z}{(\gamma+\lambda)\,(z-1)}\right)$$

$$= (1-z)^{(-a)}\left(1 + \frac{a\,z}{(2+n)\,(z-1)}\right)$$

$$\lim_{n\to\infty} (1-z)^{(-a)}\left(1 + \frac{a\,z}{(2+n)\,(z-1)}\right) = \frac{1}{(1-z)^a}$$

$$= \frac{1}{\sqrt{1-z}}$$

```
> # thus we have the rule
  hypergeom([1/2, n+1],[2+n],z):
  Limit(%, n=infinity) = evalindets(%, 'specfunc( anything, hypergeom )', f -> 1/sqrt(1 - op(3,f)));
```

$$\lim_{n\to\infty} \mathrm{hypergeom}\left(\left[\frac{1}{2}, n+1\right], [2+n], z\right) = \frac{1}{\sqrt{1-z}}$$

```
> Limit(n*Int((1-3/26*x-37/49*x^2)^n,x = 0 .. 1),n = infinity);
  ``=Limit(Sol, n=infinity);
  evalindets(%, 'specfunc( anything, hypergeom )', f -> 1/sqrt(1 - op(3,f)));
  value(%);
```

$$\lim_{n \to \infty} n \int_0^1 \left(1 - \frac{3}{26} x - \frac{37}{49} x^2\right)^n dx$$

$$= \lim_{n \to \infty} -\frac{165}{2219} \frac{n \operatorname{hypergeom}\left(\left[\frac{1}{2}, n+1\right], [2+n], \frac{634920}{4923961}\right) 165^n}{(n+1)\, 1274^n} + \frac{182}{317} \frac{n \operatorname{hypergeom}\left(\left[\frac{1}{2}, n+1\right], [2+n], \frac{100048}{100489}\right)}{n+1}$$

$$= \lim_{n \to \infty} -\frac{165\, n\, 165^n}{2071\,(n+1)\, 1274^n} + \frac{26\, n}{3\,(n+1)}$$

$$= \frac{26}{3}$$

```
>
```

Numerical testing

```
> n*Int((1-(3/26)*x-(37/49)*x^2)^n, x = 0 .. 1):
  combine(%); subs(n=100, %);
  plot(op(%));
```

$$\int_0^1 n \left(1 - \frac{3}{26} x - \frac{37}{49} x^2\right)^n dx$$

$$\int_0^1 100 \left(1 - \frac{3}{26} x - \frac{37}{49} x^2\right)^{100} dx$$



That integral is roughly = 8. Estimate, where one can cut off

```
> eps:='eps':
  eps=n*(X)^n; isolate(%, X);
  subs(X=1-3/26*x-37/49*x^2, %);
  [solve(%, x)]; #simplify(%, symbolic);
  Upper:=%[1];
```

$$eps = n\, X^n$$

$$X = \left(\frac{eps}{n}\right)^{\left(\frac{1}{n}\right)}$$

$$1-\frac{3}{26}x-\frac{37}{49}x^2=\left(\frac{eps}{n}\right)^{\left(\frac{1}{n}\right)}$$

$$\left[-\frac{147}{1924}+\frac{7\sqrt{100489-100048\left(\frac{eps}{n}\right)^{\left(\frac{1}{n}\right)}}}{1924},-\frac{147}{1924}-\frac{7\sqrt{100489-100048\left(\frac{eps}{n}\right)^{\left(\frac{1}{n}\right)}}}{1924}\right]$$

$$Upper:=-\frac{147}{1924}+\frac{7\sqrt{100489-100048\left(\frac{eps}{n}\right)^{\left(\frac{1}{n}\right)}}}{1924}$$

```
> oldDigits:=Digits:
  Digits:=3*Digits:

  kTst:=16;
  nTst:='10^kTst'; nTst:=evalf(nTst);
  eps:= 1e-12;
  ``;
  #eps:='eps';
  'Int(n*(1-3/26*x-37/49*x^2)^n,x = 0 .. 1)';
  ``='Int(n*(1-3/26*x-37/49*x^2)^n,x = 0 .. Upper)';
  subs(n=evalf(nTst), %);

  rhs(%):
  Int(op(%), epsilon=eps):
  evalf(%):
  ``=evalf[oldDigits](%);

  Digits:=oldDigits:
```

$$kTst:=16$$

$$nTst:=10^{kTst}$$

$$nTst:=0.100000000000000\ 10^{17}$$

$$eps:=0.100000000000000\ 10^{-11}$$

$$\int_0^1 n\left(1-\frac{3}{26}x-\frac{37}{49}x^2\right)^n dx$$

$$=\int_0^{Upper} n\left(1-\frac{3}{26}x-\frac{37}{49}x^2\right)^n dx$$

$$=\int_0^{0.558760649233016\ 10^{-13}} 0.100000000000000\ 10^{17}\left(1-\frac{3}{26}x-\frac{37}{49}x^2\right)^{0.100000000000000\ 10^{17}} dx$$

$$=8.666666666666570$$

```
>
>
```

Using the Laplace method

To apply the method one has to re-write the integral as $\int_0^1 f(t)\,\mathbf{e}^{(n\,\phi(t))}\,dt$:

```
> f(t)*exp(x*phi(t)): subs(x=n, %):
  'eval(%, f=1)':
  'eval(%, phi = 't -> ln(1-3/26*t-37/49*t^2)')';
  ``=simplify(%, exp) assuming 0<t,t<1,n::posint;
```

$$\left.\left.f(t)\,\mathbf{e}^{(n\,\phi(t))}\right|_{f=1}\right|_{\phi='t\to\ln\left(1-3/26\,t-\frac{37}{49}t^2\right)'}$$

$$= \left(1 - \frac{3}{26}t - \frac{37}{49}t^2\right)^n$$

The quadratics in t is strictly decreasing and has its maximum in t=0 (see first graphics), so ϕ has its maximum in c=0, since log is monotonous.

This is the case in the Malham lecture on top of page 34:

```
> F:= t -> 1;
  Phi:= t -> ln(1-3/26*t-37/49*t^2);
```

$$F := t \to 1$$

$$\Phi := t \to \ln\left(1 - \frac{3}{26}t - \frac{37}{49}t^2\right)$$

```
> c:=0;
  'Phi(c)': '%'=%;
  'D(Phi)(c)': '%'=%;
```

$$c := 0$$

$$\Phi(c) = 0$$

$$D(\Phi)(c) = \frac{-3}{26}$$

```
> -f(c)/n/D(phi)(c)*exp(n*phi(c));
  'eval(%, f=F)':
  'eval(%, phi = Phi)';
  ``=simplify(%, exp) assuming 0<t,t<1,n::posint;
```

$$-\frac{f(0)\,\mathbf{e}^{(n\,\phi(0))}}{n\,D(\phi)(0)}$$

$$\left(-\frac{f(0)\,\mathbf{e}^{(n\,\phi(0))}}{n\,D(\phi)(0)}\right)\Bigg|_{f=F}\Bigg|_{\phi=\Phi}$$

$$= \frac{26}{3\,n}$$

```
>
> Limit(n*Int((1-3/26*x-37/49*x^2)^n,x = 0 .. 1),n = infinity);
  ``='Limit(` n`*( -F(c)/n/D(Phi)(c)*exp(n*Phi(c))), n=infinity)';
  subs(` n` = n, %);
  value(%);
```

$$\lim_{n \to \infty} n \int_0^1 \left(1 - \frac{3}{26}x - \frac{37}{49}x^2\right)^n dx$$

$$= \lim_{n \to \infty} -\frac{n\,F(c)\,\mathbf{e}^{(n\,\Phi(c))}}{n\,D(\Phi)(c)}$$

$$= \lim_{n \to \infty} \frac{26}{3}$$

$$= \frac{26}{3}$$

```
>
>
```

Relating it to the numerical test

The proof for the above used formula uses the recipe (4.6) on p.33 of Step 2: approximate phi in c=a `=` `0` by a linear Taylor term, $\phi(t) \sim \phi(0) + D(\phi)(0)\,t$ and compute only in a small neighbourhood - like we did in the numerical check:

```
> Int( n*f(t)*exp(n*phi(t)), t=0..1);
  ``=Int( n*f(t)*exp(n*phi(t)), t=0..epsilon);
  'eval(%, f=F)':
  'eval(%, phi = 't -> phi(c)+D(phi)(c)*t')':
  'eval(%, phi = Phi)';
  simplify(%, exp) assuming 0<t,t<1,n::posint;
  value(%);

  Limit(rhs(%), n=infinity):
  %=value(%)  assuming 0<epsilon;
```

$$\int_0^1 n\,f(t)\,\mathbf{e}^{(n\,\phi(t))}\,dt$$

$$= \int_0^{\varepsilon} n\, f(t)\, \mathbf{e}^{(n\,\phi(t))}\, dt$$

$$\left.\left.\left.\left(= \int_0^{\varepsilon} n\, f(t)\, \mathbf{e}^{(n\,\phi(t))}\, dt \right)\right|_{f = F}\right|_{\phi = 't \to \phi(c) + D(\phi)(c)\, t'}\right|_{\phi = \Phi}$$

$$= \int_0^{\varepsilon} n\, \mathbf{e}^{\left(-\frac{3\,n\,t}{26}\right)}\, dt$$

$$= \frac{26}{3} - \frac{26}{3}\, \mathbf{e}^{\left(-\frac{3\,n\,\varepsilon}{26}\right)}$$

$$\lim_{n \to \infty} \frac{26}{3} - \frac{26}{3}\, \mathbf{e}^{\left(-\frac{3\,n\,\varepsilon}{26}\right)} = \frac{26}{3}$$

That means: the original integrand can be replaced by a simple exponential, hence allowing symbolic solutions.

In the numerical the upper bound was computed depending on n and some given error. It would be easy to see that this works as well (just do it) and thus one would like to look at the approximation errors for the integrand for the selected range:

```
> ` upper` := -147/1924+7/1924*(100489-100048*(epsilon/n)^(1/n))^(1/2);
  `----------`;
  Int( n*f(t)*exp(n*phi(t)), t=0..1);
  ``=Int( n*f(t)*exp(n*phi(t)), t=0..upper);
  'eval(%, f=F)':
  'eval(%, phi = 't -> phi(c)+D(phi)(c)*t')':
  'eval(%, phi = Phi)';
  simplify(%, exp) assuming 0<t,t<1,n::posint;
  value(%);
  eval(%, upper=` upper`);

  Limit(rhs(%), n=infinity):
  %=value(%)  assuming 0<epsilon;
```

$$\text{upper} := -\frac{147}{1924} + \frac{7\sqrt{100489 - 100048\left(\frac{\varepsilon}{n}\right)^{\left(\frac{1}{n}\right)}}}{1924}$$

```
> n = nTst;
  oldDigits:=Digits:
  Digits:=3*Digits:

  exp(-3/26*n*x)-(1-(3/26)*x-(37/49)*x^2)^n:
  n*%;
  eval(%, n=nTst):
  plot(%, x = 0 .. eval(Upper, n=nTst));
  #nTst;
  Digits:=oldDigits:
```

$$n = 0.100000000000000\ 10^{17}$$

$$n\left(\mathbf{e}^{\left(-\frac{3\,n\,x}{26}\right)} - \left(1 - \frac{3}{26}\,x - \frac{37}{49}\,x^2\right)^n\right)$$



This only looks large - the integral is very small, since it only lives in a tiny range:

```
> oldDigits:=Digits:
  Digits:=3*Digits:
  'Int(n*(exp(-3/26*n*x)-(1-3/26*x-37/49*x^2)^n), x=0 .. 1)';
  ``='Int(n*(exp(-3/26*n*x)-(1-3/26*x-37/49*x^2)^n), x=0 .. Upper)';

  op(rhs(%)):
  Int(%, epsilon=eps): # to be computed with small relative error

  eval(%, n=nTst):

  ``=evalf(%);

  Digits:=oldDigits:
```

$$\int_0^1 n\left(e^{\left(-\frac{3\,n\,x}{26}\right)}-\left(1-\frac{3}{26}x-\frac{37}{49}x^2\right)^n\right)dx$$

$$=\int_0^{Upper} n\left(e^{\left(-\frac{3\,n\,x}{26}\right)}-\left(1-\frac{3}{26}x-\frac{37}{49}x^2\right)^n\right)dx$$

$$= 0.991753590324985\ 10^{-13}$$

```
> 
> 
```

proofing, what was used from Temme

```
> Int(t^n/(1-z*t)^(1/2),t = 0 .. 1);
  ``=value(%) assuming 0<n;
```

$$\int_0^1 \frac{t^n}{\sqrt{1-z\,t}}\,dt$$

$$=\frac{\mathrm{hypergeom}\left(\left[\frac{1}{2}, n+1\right], [2+n], z\right)}{n+1}$$

```
> Int( f(t)*exp(n*phi(t)), t=0..1):
  'eval(%, f = 't -> 1/sqrt(1-z*t)')':
  'eval(%, phi=ln)';
  ``=simplify(%, exp);
```

$$\left.\left.\int_0^1 f(t)\, \mathbf{e}^{(n\,\phi(t))}\, dt\right|_{f = 't \to \frac{1}{\sqrt{1-z\,t}}'}\right|_{\phi = \ln}$$

$$= \int_0^1 \frac{t^n}{\sqrt{1-z\,t}}\, dt$$

```
> # phi has its maximum in the right boundary c=1,
  # this is the 2nd case in the Malham lecture on top of page 34:
> c:=1;
  'eval(phi(c), phi = ln)': '%'=%;
  'eval(D(phi)(c), phi = ln)': '%'=%;
```

$$c := 1$$

$$\phi(c)\Big|_{\phi = \ln} = 0$$

$$D(\phi)(c)\Big|_{\phi = \ln} = 1$$

```
> Int(t^n/(1-z*t)^(1/2),t = 0 .. 1);
  f(c)/n/D(phi)(c)*exp(n*phi(c));
  'eval(%, f='t -> 1/sqrt(1-z*t)')':
  'eval(%, phi = ln)';
  ``=simplify(%, exp) assuming 0<t,t<1,n::posint;
```

$$\int_0^1 \frac{t^n}{\sqrt{1-z\,t}}\, dt$$

$$\frac{f(1)\, \mathbf{e}^{(n\,\phi(1))}}{n\, D(\phi)(1)}$$

$$\left.\left.\frac{f(1)\, \mathbf{e}^{(n\,\phi(1))}}{n\, D(\phi)(1)}\right|_{f = 't \to \frac{1}{\sqrt{1-z\,t}}'}\right|_{\phi = \ln}$$

$$= \frac{1}{\sqrt{1-z}\; n}$$

```
> Limit( hypergeom([1/2, n+1],[2+n],z), n=infinity);
  ``=Limit( (n+1)*Int(t^n/(1-z*t)^(1/2),t = 0 .. 1), n=infinity);
  ``=Limit( (n+1)/n * 1/sqrt(1-z), n=infinity);
  value(%);
```

$$\lim_{n \to \infty} \mathrm{hypergeom}\left(\left[\frac{1}{2}, n+1\right], [2+n], z\right)$$

$$= \lim_{n \to \infty} (n+1) \int_0^1 \frac{t^n}{\sqrt{1-z\,t}}\, dt$$

$$= \lim_{n \to \infty} \frac{n+1}{n\sqrt{1-z}}$$

$$= \frac{1}{\sqrt{1-z}}$$

```
>
>
```